title: Qt QML LevelAdjust shows strange edge effects when applied to svg source
body: <p>When applying a LevelAdjust effect to an image, Qt doesn't seem to handle the edges properly.</p>
  
  <p>I have a set of svg icons, all drawn in black. I would like to change their color easily on the fly.
  I am experimenting with the LevelAdjust effect from QtGraphicalEffects 1.12. According to the doc, using minimumOutput should shift the color curve in a linear fashion: blacks will correspond to minimumOutput, white will be white, and colors in between will be a prorated version of minimumOutput.</p>
  
  <p>If I try the following:</p>
  
  <pre><code>Rectangle {
    height: 100
    width: 100
  
    Image {
      id: iconImage
      source: "qrc:/images/circles.svg"
      anchors.centerIn: parent
      width: 0.8 * parent.width
      height: width
      sourceSize.width: width
      sourceSize.height: width
      fillMode: Image.PreserveAspectFit
    }
  
    LevelAdjust {
      anchors.fill: iconImage
      source: iconImage
      minimumOutput: "#0098FE8F"
    }
  }
  </code></pre>
  
  <p>with my svg being as simple as possible:</p>
  
  <pre><code>&lt;?xml version="1.0" encoding="UTF-8"?&gt;
  &lt;svg version="1.0" viewBox="0 0 100 100" xmlns="http://www.w3.org/2000/svg"&gt;
   &lt;circle cx="50" cy="50" r="50" fill-rule="evenodd" style="paint-order:normal"/&gt;
  &lt;/svg&gt;
  </code></pre>
  
  <p>then the inside of the circle takes the expected colour. However, the edges are simply wrong. When rendering the circle, it looks like Qt does some anti-aliasing. But these extra pixels are ignored. I end up with edges which look wrong, but are also darker than the new colour (magnified result here): </p>
  
  <p><a href="https://i.stack.imgur.com/SacDG.png" rel="nofollow noreferrer"><img src="https://i.stack.imgur.com/SacDG.png" alt="resulting image"></a></p>
  
  <p>I would have expected their colour to be shifted as needed too.
  The result is the same on windows, and Android. Am I doing anything wrong?</p>
  
  <hr>
  
  <p><strong>Edit on 2019/09/05:</strong></p>
  
  <p>Based on the accepted answer and suggestion below, I created the following ColoredImage.qml component:</p>
  
  <pre><code>import QtQuick 2.0
  import QtGraphicalEffects 1.12
  
  Item {
    id: imageWrapper
    property bool colored: true
    property string color: "#FFFF0000"
    // Replicate some Image properties:
    property string source
    property var fillMode: Image.PreserveAspectFit
    property var status: image.status
    property real progress: image.progress
  
    Image {
      id: image
      source: imageWrapper.source
      anchors.fill: parent
      sourceSize.width: width
      sourceSize.height: height
      fillMode: imageWrapper.fillMode
      visible: !imageWrapper.colored
    }
  
    ColorOverlay {
      id: overlay
      anchors.fill: image
      source: image
      color: imageWrapper.color
      visible: imageWrapper.colored
    }
  }
  </code></pre>
  
  <p>Thanks again!</p>